
10

OPERATORY RELACYJNE
I LOGICZNE

W tym rozdziale dowiesz się:
 Co to są operatory relacyjne

i jak je wykorzystywać
 Co to są operatory logiczne

i jak ich używać

 Jak stosować operatory poziomu
bitowego

 Jakie priorytety mają operatory
relacyjne i logiczne

ednym z podstawowych zadań niemal każdego programu komputerowego
jest zdolność podejmowania decyzji na podstawie dostarczonych danych.
Kluczowe znaczenie dla tej zdolności mają operatory relacyjne i logiczne.

W programach służą one do podejmowania decyzji na podstawie porównania jednych
wartości z innymi. W tym rozdziale zajmiemy się takimi operatorami dostępnymi
w języku Visual Basic .NET.

Operatory relacyjne
Operatory relacyjne (relational operators) służą do porównywania wartości. Tabela 10.1
przedstawia ich zestawienie.

Wszystkie operatory relacyjne są binarne, to znaczy, że (jak przypominamy sobie
z poprzednich rozdziałów) wymagają dwóch argumentów. Ogólna składnia grupy
operatorów relacyjnych wygląda następująco:

Argument1 OperatorRelacyjny Argument2

J

TABELA 10.1. Operatory relacyjne języka Visual Basic .NET

Operator relacyjny Znaczenie
= Równe

<> Różne

< Mniejszy od

> Większy od

<= Mniejszy lub równy

>= Większy lub równy

Gdy w jakimś wyrażeniu wykorzystywany jest operator relacyjny, mówi się, że
pomiędzy dwiema wartościami (to jest argumentami) wykonywany jest test relacyjny
(relational test). Test taki może dać tylko jeden z dwóch wyników: logiczną prawdę
(wartość True) lub logiczny fałsz (wartość False).

Proste wyrażenie If-Then-Else

Testy relacyjne są często wykonywane przy użyciu wyrażeń If-Then-Else. Mimo
iż używaliśmy ich już w poprzednich rozdziałach, nie zapoznaliśmy się do tej pory
z ich składnią:

If Wyrażenie1 Then
  Wykonaj, jeśli Wyrażenie1 jest prawdziwe
Else
  Wykonaj, jeśli Wyrażenie1 nie jest prawdziwe
End If

Blok wyrażenia If rozpoczyna się od słowa kluczowego If, a kończy instrukcją End
If. Wyrażenie1 jest zwykle testem logicznym, którego wynikiem może być jedna
z wartości: True lub False. Jeśli w wyniku przeprowadzenia testu Wyrażenie1 otrzy-
mamy wartość True, wykonywana jest druga linia bloku If. Jeśli otrzymamy wartość
False, wykonywana jest linia znajdująca się po słowie kluczowym Else.

Porada dla programistów
Wyrażenia do wykonania w wyniku przeprowadzenia testu logicznego w bloku
If-Then-Else, zgodnie z obowiązującą konwencją, są wcięte o jeden tabula-
tor. Więcej na temat wyrażenia If-Then-Else powiemy w rozdziale 11.

Napiszemy teraz prosty program, który pozwoli przećwiczyć zarówno składnię If,
jak i operatory relacyjne. Rozpocznijmy nowy projekt i nazwijmy go RelationalO-
perators (Operatory relacyjne). Dodajmy do jego formularza pola tekstowe i przyciski
pokazane na rysunku 10.1. (Można wykorzystać program napisany w rozdziale 9.
w celu testowania operatorów arytmetycznych).

RYSUNEK 10.1.
Program
do wykonywania
testów relacyjnych

Na rysunku 10.1 widzimy pola tekstowe przeznaczone na oba argumenty: txtAr-
gument1 i txtArgument2, oraz trzecie pole tekstowe służące do pokazania całego
sprawdzanego wyrażenia. W polu o nazwie txtExpression widać będzie całe
Wyrażenie1 z przedstawionej przed chwilą składni bloku If-Then-Else. U dołu
formularza widocznego na rysunku 10.1 znajduje się pole grupy zawierające dwa pola
wyboru. Będą one służyć do prezentacji wyniku przeprowadzanego testu logicznego.

Kod programu jest bardzo prosty. Pole grupy o nazwie Operator zawiera sześć
przycisków opcji, po jednym dla każdego operatora relacyjnego. Poniżej przedsta-
wiony jest kod związany z pierwszym przyciskiem opcji:

Private Sub RadioButton1_CheckedChanged(ByVal sender As System.Object, _
    ByVal e As System.EventArgs) Handles RadioButton1.CheckedChanged
  WhichOne = 1
End Sub

W wyniku kliknięcia określonego przycisku opcji zmienna WhichOne przyjmuje war-
tość odpowiadającą temu przyciskowi. WhichOne jest zmienną typu Integer zdefi-
niowaną z zasięgiem modułowym.

Po wpisaniu argumentów do pól tekstowych i wybraniu operatora relacyjnego użyt-
kownik klika przycisk Test. Pełny kod obsługi zdarzenia Click() obiektu btnTest
jest pokazany na listingu 10.1.

LISTING 10.1. Kod obsługi zdarzenia Click() obiektu btnTest

Private Sub btnTest_Click(ByVal sender As System.Object, _
          ByVal e As System.EventArgs) _
          Handles btnTest.Click
  ckbTrue.Checked = False   ' Rozpoczynamy od ustawienia wartości False
  ckbFalse.Checked = False

  Select Case WhichOne
    Case 1
      txtExpression.Text = txtArgument1.Text & " = " & txtArgument2.Text
      If txtArgument1.Text = txtArgument2.Text Then ' A1 równa się A2
        ckbTrue.Checked = True
      Else
        ckbFalse.Checked = True
      End If
    Case 2
      txtExpression.Text = txtArgument1.Text & " <> " & txtArgument2.Text
      If txtArgument1.Text <> txtArgument2.Text Then 'A1 jest różny od A2
        ckbTrue.Checked = True
      Else
        ckbFalse.Checked = True
      End If
    Case 3
      txtExpression.Text = txtArgument1.Text & " > " & txtArgument2.Text
      If txtArgument1.Text > txtArgument2.Text Then 'A1 jest większy od A2
        ckbTrue.Checked = True
      Else
        ckbFalse.Checked = True
      End If
    Case 4
      txtExpression.Text = txtArgument1.Text & " < " & txtArgument2.Text
      If txtArgument1.Text < txtArgument2.Text Then 'A1 jest mniejszy
                                                    'od A2
        ckbTrue.Checked = True
      Else
        ckbFalse.Checked = True
      End If
    Case 5
      txtExpression.Text = txtArgument1.Text & " >= " & txtArgument2.Text
      If txtArgument1.Text >= txtArgument2.Text Then 'A1 jest większy
                                                     'lub równy A2
        ckbTrue.Checked = True
      Else
        ckbFalse.Checked = True
      End If
    Case 6
      txtExpression.Text = txtArgument1.Text & " <= " & txtArgument2.Text
      If txtArgument1.Text <= txtArgument2.Text Then 'A1 jest mniejszy
                                                     'lub równy A2
        ckbTrue.Checked = True
      Else
        ckbFalse.Checked = True
      End If
  End Select
End Sub

Kod procedury rozpoczyna się od nadania obu polom wyboru wartości False. Na-
stępnie wykorzystywany jest blok Select Case służący do wybrania właściwego
fragmentu kodu na podstawie wartości zmiennej WhichOne. W każdym z wariantów

wyrażenia Select budujemy ciąg znaków txtExpression poprzez złączenie argu-
mentów z pól txtArgument1 oraz txtArgument2. Pomiędzy nimi umieszczana jest
stała tekstowa reprezentująca wybrany operator. Na końcu, za pomocą prostej instruk-
cji If decydujemy, czy ma być zaznaczone pole wyboru True, czy False. Wszystkie
bloki Case zawierają kod działający na tej samej zasadzie.

Przykładowe wykonanie naszego programu widzimy na rysunku 10.2.

RYSUNEK 10.2.
Przykładowe
wykonanie programu
RelationalOperators

Warto zauważyć, że operatory relacyjne można wykorzystywać również do porów-
nywania danych tekstowych. Przykład pokazuje, że wyrażenie testowe

"A" > "B"

daje logiczną wartość False. Dlaczego wyrażenie to jest nieprawdziwe? Przypomnijmy
sobie, że porównywanie znaków odbywa się na podstawie liczbowych wartości kodów
liter „A” i „B”. Kod litery „A” wynosi 65, a kod litery „B” wynosi 66, zatem całe wyra-
żenie nie jest prawdziwe. Oczywiście, jako argumenty do porównania można podać
wartości liczbowe.

Warto przetestować działanie programu, aby dobrze zapoznać się z funkcjonowaniem
operatorów relacyjnych. Do testów powinniśmy wykorzystać zarówno dane licz-
bowe, jak i tekstowe. Zmienne tekstowe mogą przy tym zawierać więcej niż jeden
znak. Jaki na przykład będzie wynik porównania napisów „JOYCE” i „Joyce” i jak
można go uzasadnić?

Operatory logiczne
Operator logiczny (logical operator) służy do porównywania argumentów typu Bo-
olean i zwraca wartość True lub False (wynik typu Boolean). W tabeli 10.2 przed-
stawiona jest lista operatorów logicznych dostępnych w języku Visual Basic .NET.

Jak widać w tabeli 10.2, wszystkie operatory logiczne z wyjątkiem negacji są binarne.
W kolejnym podrozdziale omówimy dokładnie każdy z nich.

TABELA 10.2. Operatory logiczne języka Visual Basic .NET

Operator Logiczny Typ Znaczenie
And Binarny Logiczne „i”

Or Binarny Logiczne „lub”

Xor Binarny Logiczne „lub” wyłączające

Not Unarny Logiczna negacja

AndAlso Binarny Warunkowe „i”

OrElse Binarny Warunkowe „lub”

Operator logiczny And

Operator logiczny And (logiczne „i”) porównuje dwa argumenty i zwraca logiczną
wartość True wtedy i tylko wtedy, gdy oba argumenty mają logiczną wartość True.
Relację tę można zobaczyć w tabeli prawdy operatora And. Tabela prawdy (truth table)
pokazuje listę możliwych kombinacji wartości argumentów i wynik wykonania na
nich operacji logicznego dodawania. Takie zestawienie przedstawia tabela 10.3.

TABELA 10.3. Tabela prawdy operatora logicznego And

Argument1 Argument2 Wynik
True True True

True False False

False True False

False False False

Zwróćmy uwagę, że wynik True dla operatora logicznego And uzyskujemy tylko
wtedy, gdy oba argumenty mają wartość True. Każda inna kombinacja daje wynik
False. Przyjrzyjmy się następującemu wyrażeniu:

Result = 10 > 5 And 100 > 60

Jeśli wykonamy wszystkie testy relacyjne, otrzymamy:
Result = 10 > 5 And 100 > 60
Result = True And True  ' kolumny 1. i 2. w wierszu 1. tabeli 10.3
Result = True           ' wiersz 1., kolumna 3. tabeli 10.3

Ze względu na to, że 10 jest większe od 5, pierwszy test relacyjny daje wynik True.
Również 100 jest większe od 60, więc drugi test także daje wynik True. Z tabeli 10.3
dla wartości True obu argumentów odczytujemy wynik True.

Spróbujmy zmodyfikować nasze wyrażenie, odwracając drugi operator relacyjny:
Result = 10 > 5 And 100 < 60
Result = True And False  ' kolumny 1. i 2. w wierszu 3. tabeli 10.3
Result = False           ' wiersz 3., kolumna 3. tabeli 10.3

Oto jeszcze kilka przykładów wraz z wynikami pokazanymi jako komentarze na
końcu linii:

Result = 10 = 5 And 100 > 60              ' False
Result = 10 >= 5 And 100 <> 60            ' True
Result = 10 <> 5 And 100 <> 60            ' True
Result = 10 > 5 And 100 <= 60             ' False
Result = 10 > 5 And 100 > 60 And 20 < 80  ' True
Result = 10 > 5 And 100 < 60 And 20 < 80  ' False

Proszę przeanalizować wszystkie przykłady, aby nabrać przekonania o poprawności
pokazanych wyników. W ostatnich dwóch przykładach widać, że w wyrażeniu może
się znajdować więcej niż jeden operator logicznego „i”.

Operator logiczny Or

Operator logicznego „lub” porównuje dwa argumenty i zwraca wynik True, jeśli któ-
rykolwiek z nich ma wartość True. Poniżej przedstawiona jest jego tabela prawdy.

TABELA 10.4. Tabela prawdy operatora Or

Argument1 Argument2 Wynik

True True True

True False True

False True True

False False False

Test logiczny Or daje wynik True zawsze wtedy, gdy którykolwiek z argumentów ma
wartość True. Wynik False uzyskujemy tylko wtedy, gdy oba argumenty mają war-
tość False. Użyjmy tych samych przykładów, co poprzednio, wykonując na nich
tym razem logiczny test „lub”:

Result = 10 = 5 Or 100 > 60               ' True
Result = 10 >= 5 Or 100 <> 60             ' True
Result = 10 <> 5 Or 100 <> 60             ' True
Result = 10 > 5 Or 100 <= 60              ' True
Result = 10 > 5 Or 100 > 60 Or 20 < 80    ' True
Result = 10 > 5 Or 100 < 60 Or 20 < 80    ' True

Ze względu na to, że w każdym przykładzie przynajmniej jedno wyrażenie jest praw-
dziwe, w każdym uzyskujemy wynik True. Cennym ćwiczeniem może być porów-
nanie wyników uzyskanych przy użyciu operatorów And i Or, gdyż ułatwi to zrozu-
mienie różnicy między nimi.

Operator logiczny Xor

Operator Xor wykonuje test logiczny wyłączającego „lub”. Zwraca on wartość True
wtedy i tylko wtedy, gdy tylko jeden z argumentów ma wartość True. Gdy oba ar-
gumenty mają wartość True lub oba mają wartość False, uzyskujemy wynik False.
Poniżej przedstawiona jest tabela prawdy logicznego operatora Xor.

TABELA 10.5. Tabela prawdy logicznego operatora Xor

Argument1 Argument2 Wynik

True True False

True False True

False True True

False False False

Użyjmy znowu tych samych przykładów, stosując w nich operator logiczny Xor:
Result = 10 = 5 Xor 100 > 60                ' True
Result = 10 >= 5 Xor 100 <> 60              ' False
Result = 10 <> 5 Xor 100 <> 60              ' False
Result = 10 > 5 Xor 100 <= 60               ' True
Result = 10 > 5 Xor 100 > 60 Xor 20 < 80    ' True
Result = 10 > 5 Xor 100 < 60 Xor 20 < 80    ' False

Pierwsze cztery przykłady są łatwe do zrozumienia, ale ostatnie dwa wymagają wy-
jaśnień. Przyjrzyjmy się pierwszemu z nich:

Result = 10 > 5 Xor 100 > 60 Xor 20 < 80
Result = True Xor True Xor True
Result = False Xor True
Result = True                               ' True

Główna trudność pojawia się w kroku 2. Ponieważ rozwiązujemy wyrażenie z dwoma
operatorami Xor, obie operacje mają taki sam priorytet. Dlatego też trzeba spojrzeć
na ten krok tak, jakby był zapisany w następujący sposób:

Result = (True Xor True) Xor True           ' True

W tabeli 10.5 widzimy, że gdy oba argumenty mają wartość True, test Xor daje wynik
False. W ostatnim kroku False Xor True daje wynik True.

Drugie złożone wyrażenie jest obliczane w taki sposób:
Result = 10 > 5 Xor 100 < 60 Xor 20 < 80
Result = (True Xor False) Xor True
Result = True Xor True
Result = False                              ' False

Ostatecznym wynikiem jest wartość False, gdyż w ostatnim kroku wykonujemy test
na dwóch wyrażeniach o wartości True.

Prawdopodobnie nie będziemy zbyt często wykorzystywać operatora Xor. Jest on
jednak bardzo przydatny w niektórych zastosowaniach, na przykład przy operowaniu
na obrazach.

Operacje poziomu bitowego wykonywane
za pomocą operatorów logicznych

Czasami pojawia się konieczność zejścia na poziom bitów pewnych danych. Przy-
kłady zastosowań takiego postępowania można znaleźć w dekodowaniu zakodo-
wanych danych, w oprogramowaniu komunikacyjnym i graficznym. Aby przyjrzeć
się operacjom na poziomie bitowym, wykorzystamy kod programu pokazanego
na rysunku 10.1. Musimy uzupełnić procedurę obsługi zdarzenia Click() obiektu
btnTest zgodnie z listingiem 10.2. Pokazanych jest tu tylko kilka pierwszych linii,
ponieważ reszta pozostała niezmieniona w stosunku do listingu 10.1.

LISTING 10.2. Częściowy listing kodu procedury obsługi zdarzenia Click() obiektu btnTest

Private Sub btnTest_Click(ByVal sender As System.Object, ByVal e As _
             System.EventArgs) Handles btnTest.Click
  ckbTrue.Checked = False   ' Rozpoczynamy od ustawienia wartości False
  ckbFalse.Checked = False
  Dim Result As Boolean    ' Tu jest nowa linia kodu!

  Result = CBool(5 And 4)      ' Ostatnia linia nowego kodu!

  Select Case WhichOne
    Case 1
      ' Pozostała część kodu procedury

W powyższym listingu zwróćmy uwagę na dwie nowe linie, w których zdefiniowana
jest zmienna o nazwie Result i typie danych Boolean. W kolejnej linii wykonywany
jest test logiczny And na wartościach 5 i 4. Skompilujmy program, ale poczekajmy
jeszcze z jego uruchamianiem.

Ustawianie punktu przerwania w debugerze

Widząc w oknie Form kod procedury obsługi zdarzenia Click() obiektu btnTest,
kliknijmy lewy margines linii, w której znajduje się polecenie Select Case. Powinna
się pojawić brązowoczerwona linia podobna do pokazanej na rysunku 10.3. Nosi
ona nazwę linii punktu przerwania (breakpoint line) i oznacza, że w tym miejscu został
ustanowiony punkt przerwania programu.

Punkt przerwania (breakpoint) jest właściwością debugera Visual Basic .NET umożli-
wiającą zatrzymanie wykonywania programu na wybranej linii. Po jej osiągnięciu
działanie programu zostaje wstrzymane. Można wtedy skontrolować wartości wszel-
kich zmiennych istniejących w danym momencie.

RYSUNEK 10.3.
Ustawianie punktu
przerwania
w procedurze obsługi
zdarzenia Click()
obiektu btnTest

Po ustawieniu punktu przerwania i skompilowaniu programu należy nacisnąć klawisz
<F5>. (Przypomnijmy, że naciśnięcie <F5> jest jednym ze sposobów uruchamiania
programu). Po tym należy kliknąć przycisk Test. Na rysunku 10.4 pokazany jest stan
programu po osiągnięciu punktu przerwania.

RYSUNEK 10.4.
Osiągnięcie punktu
przerwania podczas
wykonywania procedury

W momencie wstrzymania wykonywania programu w wyznaczonej linii zmienia
ona swój kolor z brązowoczerwonego na żółty, aby wskazać miejsce, w którym znaj-
duje się sterowanie. W lewym dolnym rogu rysunku 10.4 widzimy zakładki zatytuło-
wane: Autos, Locals i Watch. Kliknijmy zakładkę Locals. Ekran powinien wyglądać

podobnie do tego z rysunku 10.5. (Możemy zatrzymać działanie debugera, wy-
bierając z menu polecenie Debug, Stop Debugging lub naciskając niebieski kwa-
dracik na pasku narzędzi).

RYSUNEK 10.5.
Okno zakładki Locals
dla procedury obsługi
zdarzenia Click()
obiektu btnTest

Na rysunku 10.5 widzimy, że w oknie Locals pokazane są zmienne o zasięgu lokal-
nym. Jest wśród nich zmienna Result mająca wartość True. Tak więc test logiczny
And przeprowadzony na wartościach 5 i 4 daje wynik True. Dlaczego wynik jest wła-
śnie taki? Wartości 5 i 4 są niezerowe, przez co są widziane jako argumenty o war-
tości logicznej True, stąd wynik testu ma wartość typu Boolean True.

Teraz zmienimy definicję zmiennej Result z typu Boolean na Integer:
Dim Result As Integer

Następnie skompilujemy i uruchomimy program. Wynik powinien wyglądać podobnie
do przedstawionego na rysunku 10.6.

Result ma teraz wartość 4 zamiast True. Wynika to ze zdefiniowania tej zmiennej
jako typu danych Integer, a nie Boolean. Gdy była ona typu Boolean, mogła przyj-
mować tylko jedną z dwóch wartości: True lub False. Teraz Result jest w stanie
przyjąć wynik operacji And wykonanej na poziomie bitowym na wartościach 5 i 4.
Na czym polega taka operacja?

Operacja And poziomu bitowego polega na porównaniu dwóch wartości bit po bicie,
czyli wykonaniu logicznego testu And ich poszczególnych bitów. Oto binarne repre-
zentacje argumentów:

5 = 00000101
4 = 00000100

RYSUNEK 10.6.
Zmienna Result
przedstawiająca wynik
operacji And wykonanej
na poziomie bitowym

Pamiętamy, że pierwszym bitem liczby jest ten, który znajduje się najbardziej po
prawej stronie. W przypadku liczby 5 bit ten ma wartość 1. W przypadku liczby
4 pierwszy bit ma wartość 0. Z rozdziału 4. przypominamy sobie, że jedynkę można
traktować jako logiczną wartość True, a zero jako logiczną wartość False. Tak więc
pierwszy bit liczby 5 ma wartość True, a pierwszy bit liczby 4 — wartość False.
Spójrzmy teraz na tabelę 10.3 przedstawiającą wyniki testu logicznego And, gdzie True
And False daje wynik False, czyli 0.

Jeśli zastosujemy operator And dla wszystkich bitów obu liczb, uzyskamy następujący
wynik:

5 = 00000101
4 = 00000100
------------
    00000100 = operacja poziomu bitowego And dla argumentów 5 i 4

Zauważmy, że w wyniku przeprowadzenia operacji And na wszystkich bitach tylko na
pozycji 3. uzyskujemy wynik 1. Jest tak dlatego, że na pozycji 3. oba bity mają wartość
logiczną True. Jeśli przekonwertujemy wynik na liczbę dziesiętną, otrzymamy wartość 4.
Dokładnie taką wartość zmiennej Result widzimy w oknie Locals na rysunku 10.6.

Zobaczmy, co się będzie działo dla innych wartości. Spróbujmy zmienić argumenty
na 16 i 4, zrekompilujmy program i obejrzyjmy wyniki. Oto one:

16 = 00010000
 4 = 00000100
-------------
     00000000 = operacja poziomu bitowego And dla argumentów 16 i 4

W tym przypadku, ze względu na brak wartości 1 jednocześnie dla bitów obu liczb
na którejkolwiek z pozycji, wynik wynosi 0.

Operacje poziomu bitowego Or i Xor

Na poziomie bitowym można też wykorzystywać funkcje Or i Xor. Aby zobaczyć
ich działanie, zmieńmy instrukcję z operatorem And z listingu 10.2 na następującą:

Result = 5 Or 4

Po skompilowaniu i uruchomieniu programu zobaczymy wynik. Na podstawie tabeli
10.4 możemy na poziomie poszczególnych bitów oczekiwać następujących wyników:

5 = 00000101
4 = 00000100
------------
    00000101 = operacja poziomu bitowego Or dla argumentów 5 i 4

Jest to wartość 5. Logiczna operacja Or daje wynik True, gdy którykolwiek z argu-
mentów ma wartość True. Jaki będzie wynik operacji 16 Or 4?

16 = 00010000
 4 = 00000100
-------------
     00010100 = operacja poziomu bitowego Or dla argumentów 16 i 4

Jeśli przekonwertujemy tę wartość binarną, otrzymamy liczbę 20.

Teraz wykonajmy operację Xor na liczbach 5 i 4, odczytując wynik z tabeli 10.5:
5 = 00000101
4 = 00000100
------------
    00000001 = operacja poziomu bitowego Xor dla argumentów 5 i 4

Ze względu na to, że Xor zwraca logiczną wartość True tylko wtedy, gdy oba bity
różnią się, tylko dla pierwszego bitu otrzymujemy wynik True. W systemie dzie-
siętnym jest to liczba 1. Jaki będzie wynik dla argumentów 16 i 4?

16 = 00010000
 4 = 00000100
-------------
     00010100 = operacja poziomu bitowego Xor dla argumentów 16 i 4

Wynik wynosi 20.

Raczej rzadko będziemy korzystać w swoich programach z operacji na poziomie
bitowym. Są one przydatne w pewnych zastosowaniach, takich jak oprogramowanie
telekomunikacyjne (na przykład oprogramowanie modemów), przetwarzanie obrazów
i algorytmy kodowania czy dekodowania.

Priorytety operatorów
Omówiliśmy do tej pory operatory arytmetyczne, relacyjne i logiczne dostępne
w języku Visual Basic .NET. W skomplikowanych wyrażeniach mogą się znaleźć ich
wszystkie trzy typy jednocześnie. W jaki sposób Visual Basic .NET określa kolejność

wykonywania poszczególnych operacji w takich złożonych instrukcjach? O kolejności
decydują priorytety operatorów.

W tabeli 10.6 przedstawiona jest kolejność wykonywania wszystkich operatorów
języka Visual Basic .NET, począwszy od posiadających najwyższy priorytet do tych
o coraz niższych priorytetach. (Tabela 10.6 zawiera także kilka operatorów, które
będą omówione dopiero w dalszych rozdziałach).

TABELA 10.6. Priorytety operatorów w języku Visual Basic .NET

Operator Nazwa Typ
^ Potęga Arytmetyczny

- Negacja Arytmetyczny

*, / Mnożenie, dzielenie Arytmetyczny

\ Dzielenie całkowitoliczbowe Arytmetyczny

Mod Modulo Arytmetyczny

+, - Dodawanie, odejmowanie Arytmetyczny

& Łączenie tekstów Łączenie tekstów

= Równość Relacyjny

<> Nierówność Relacyjny

<, > Mniejszy, większy Relacyjny

>= Większy lub równy Relacyjny

<= Mniejszy lub równy Relacyjny

Like Porównywanie tekstów Porównywanie tekstów

Is Porównywanie obiektów Porównywanie obiektów

TypeOf...Is Porównywanie typów Porównywanie typów

Not Negacja Logiczny

And, AndAlso „i”, „także” Logiczny

Or, Xor,
OrElse

„lub”, wyłączające „lub”, „w przeciwnym
razie”

Logiczny

= Przypisanie Przypisanie

Ogólna zasada jest taka, że najwyższy priorytet mają operatory arytmetyczne, na-
stępnie relacyjne, a później logiczne. Chyba warto zagiąć narożnik tej strony, aby ła-
two znaleźć powyższą tabelę, gdy zaczniemy omawiać bardziej złożone przykłady
programów.

Należy również pamiętać, że kolejność wykonywania operatorów można zmienić,
stosując nawiasy (patrz rozdział 9).

Podsumowanie
W tym rozdziale powiedzieliśmy, w jaki sposób w programach można używać ope-
ratorów relacyjnych i logicznych. Zobaczyliśmy też, jak można posługiwać się tabelami
prawdy do sprawdzania wyników operacji logicznych. Omówiliśmy wpływ prio-
rytetów na oba typy operatorów. Na końcu nauczyliśmy się stosowania operatorów
logicznych do wykonywania operacji na poziomie bitowym. Mimo iż niektórych
omówionych tu technik nie będziemy w tej chwili potrzebować w pisanych progra-
mach, warto zachować w pamięci te informacje, gdyż w przyszłości może się okazać,
że pomogą nam one elegancko rozwiązać jakiś trudny problem.

Pytania kontrolne

Porada dla programistów
Testując krótkie programy przedstawione w tej części, można wpisać ich kod
do procedury obsługi zdarzenia Load() formularza. Nie potrzeba też używać pól
tekstowych ani żadnych innych formantów. W celach testowych wystarczy umieścić
punkt przerwania w linii End Sub tej procedury. Można, oczywiście, analizować
wartości wszelkich zmiennych, umieszczając nad nimi kursor.

 1. Do czego służą operatory relacyjne?

O: Służą one do porównywania wartości. Ze względu na to, że porównywane są
zawsze dwie wartości, są to operatory binarne wymagające dwóch argumentów.
Operatory te są podstawą wszelkiego podejmowania decyzji w programach na
podstawie wartości danych.

 2. Przypuśćmy, że mamy zmienną o nazwie Gender (płeć), przyjmującą wartość 0
dla kobiet i 1 dla mężczyzn. Jak napisać kod programu sprawdzającego wartość
zmiennej Gender i na tej podstawie wyświetlającego napis „Mężczyzna” lub
„Kobieta” w oknie komunikatu?

O: Kod może wyglądać następująco:
Dim Gender As Integer ' Płeć: 0 — Kobieta, 1 - Mężczyzna

Gender = 0

If Gender = 0 Then
  MessageBox.Show("Kobieta")   ' Wyświetl komunikat, jeśli to kobieta
Else
  MessageBox.Show("Mężczyzna") ' Wyświetl komunikat, jeśli to mężczyzna
End If

W tym przykładzie wykonujemy test If przy użyciu operatora równości, na pod-
stawie wyniku określamy, którą płeć wyświetlić, wywołując metodę Message-
Box().

 3. Jaką wartość przyjmie zmienna Result po wykonaniu następującej instrukcji?
Result = 10 > 5

O: Można odpowiedzieć na to pytanie, wykonując następujący kod:
Dim Result As Integer
Result = 10 > 5
MessageBox.Show(CStr(Result))

Jaką wartość wyniku przewidujemy i dlaczego właśnie taką? Jeśli po urucho-
mieniu programu otrzymamy wynik inny od oczekiwanego, spróbujmy ustalić,
jaka jest tego przyczyna. (Wskazówka: w języku Visual Basic .NET logiczna war-
tość True jest reprezentowana przez liczbę –1, a wartość False — liczbę 0).

 4. Jaką wartość przyjmie zmienna Result po wykonaniu następującej instrukcji?
Result = (10 > 5) And (5 > 7)

O: Odpowiedź można sprawdzić za pomocą następującego kodu:
Dim Result As Integer
Result = (10 > 5) And (5 > 7)
MessageBox.Show(CStr(Result))

Gdyby uzyskana odpowiedź była inna od oczekiwanej, gdzie popełniliśmy błąd?

 5. Jaką wartość przyjmie zmienna Result po wykonaniu następującej instrukcji?
Result = (10 > 5) Or (5 > 7)

O: Odpowiedź można sprawdzić za pomocą następującego kodu:
Dim Result As Integer
Result = (10 > 5) Or (5 > 7)
MessageBox.Show(CStr(Result))

Powinniśmy umieć wyjaśnić, dlaczego zmienna Result przyjmuje właśnie taką
wartość. Ważne jest to, aby odróżnić tę odpowiedź od odpowiedzi na pytanie kon-
trolne numer 4.

 6. Jaką wartość przyjmie zmienna Result po wykonaniu następującej instrukcji?
Result = Not 3

O: Odpowiedź można sprawdzić za pomocą następującego kodu:
Dim Result As Integer
i = 3
Result = Not 3
MessageBox.Show(CStr(Result))

Wynikiem unarnej operacji Not jest bitowe uzupełnienie negowanego wyrażenia
— w tym przypadku liczby 3. Oto binarna reprezentacja liczby 3:

00000011

Bitowe uzupełnienie polega na odwróceniu każdego bitu wyrażenia, tak więc po
wykonaniu operacji Not otrzymamy w zapisie bitowym:

11111100

Zauważmy, że wszystkie jedynki stają się zerami i odwrotnie. Jednak ze względu
na włączenie bitu znaku wynik jest liczbą ujemną. Ze względu na sposób inter-
pretacji bitu znaku odpowiedź brzmi więc –4.

 7. Wyobraźmy sobie funkcję o nazwie IsItOdd() zwracającą wynik niezerowy,
gdy przekazana do niej wartość jest nieparzysta, oraz wynik 0, gdy wartość ta
jest parzysta. Co zostanie wyświetlone w oknie komunikatu po uruchomieniu
programu?

Private Sub Form1_Load(ByVal sender As System.Object, ByVal e As _
            System.EventArgs) Handles MyBase.Load
  Dim Result As Integer, TestValue As Integer
  TestValue = 55 < 100 ' liczbowa wartość testu logicznego

  If IsItOdd(TestValue) Then
    MessageBox.Show("Witaj")
  Else
    MessageBox.Show("Do zobaczenia")
  End If
End Sub

Private Function IsItOdd(ByVal num As Integer) As Integer
  num = num Mod 2
  Return num
End Function

O: Interesującą częścią kodu jest test If. Pierwszy test sprawdzający, czy 55 jest mniej-
sze od 100 daje logiczny wynik True, czyli przypisanie do zmiennej TestValue
wartości –1. Funkcja IsItOdd() wywoływana jest z parametrem –1, czyli liczbą
nieparzystą (ujemną). Funkcja zwraca zatem wynik –1. Po wyjściu z funkcji testu-
jemy za pomocą instrukcji If zwracaną wartość –1, a ponieważ każda liczba
niezerowa jest uznawana za logiczną wartość True, program wyświetla w oknie
komunikatu napis „Witaj”.

 8. Wartość bieżącą przepływu środków pieniężnych można obliczyć za pomocą
następującego równania:

PV = Dochód / (1.0 + Oprocentowanie) ^ Lata

Jeśli na przykład Dochód wynosi 100 dolarów, oprocentowanie 5% a liczba lat 2,
wartość bieżąca wynosi 90,70. Oznacza to, że jeśli zainwestujemy 90,70 na 2 lata
w aktywa o stopie zwrotu 5%, po upływie tych dwóch lat będziemy mieć 100.
Proszę napisać kod do obliczenia wartości bieżącej dla 200 dolarów zainwesto-
wanych na 3 lata ze stopą zwrotu 5%.

O: Oto kod:
Dim Value As Double
Value = 200 / (1.05) ^ 3

Wynik wynosi 172,77 dolarów. Czy nawiasy są potrzebne? Dlaczego? Można
spróbować uogólnić ten program, używając pól tekstowych do wprowadzania
oprocentowania, kwoty i liczby lat oraz wyświetlania wyniku.


